

Zpravodaj Československého sdružení uživatelů TeXu

Peter Wilson

Mělo by to fungovat VIII – Kresby

Zpravodaj Československého sdružení uživatelů TeXu, Vol. 29 (2019), No. 1-4, 92–104

Persistent URL: <http://dml.cz/dmlcz/150180>

Terms of use:

© Československé sdružení uživatelů TeXu, 2019

Institute of Mathematics of the Czech Academy of Sciences provides access to digitized documents strictly for personal use. Each copy of any part of this document must contain these *Terms of use*.



This document has been digitized, optimized for electronic delivery and stamped with digital signature within the project *DML-CZ*:
The Czech Digital Mathematics Library <http://dml.cz>

V článku jsou diskutovány některé otázky vkládání METAPOSTových obrázků do (pdf)L^AT_EXového dokumentu. Dále se článek zaměřuje na kreslení obrazců zvaných spidrony a na manipulace s nimi.

Klíčová slova: L^AT_EX, METAPOST, spidron

Calm was the day and through the trembling air
Sweet-breathing Zephyrus did softly play—
A gentle spirit, that lightly did delay
Hot Titan's beams, which then did glister fair.

Prothalamion

EDMUND SPENSER

Cílem tohoto seriálu je ukázat čtenáři krátké kousky kódu, které mohou vyřešit některé z jeho problémů. Doufám, že situaci ještě více nezkomplikuji v důsledku mých chyb. Opravy, poznámky a návrhy na změny budou vždy vítány.

Tento díl se týká (L^A)T_EXu pouze okrajově.

Nothing in India is identifiable, the mere asking of
a question causes it to disappear or to merge into
something else.

A Passage to India

E. M. FORSTER

1. Repríza

V reakci na předchozí díl [2] mi profesor Klaus Lagally ukázal další způsob, jakým lze odstranit přebytečný znak na konci příkazu. V článku jsem ukazoval krátký kód, který fungoval podobným způsobem jako L^AT_EXová makra s hvězdičkou, přičemž namísto hvězdičky jsem používal otazník. Problémem bylo zjistit, zda se za názvem makra vyskytuje otazník, a v závislosti na tom něco vykonat. Dále bylo třeba případný otazník na vstupu odstranit. V článku jsem ukázal následující řešení.

Z anglického originálu *Glisterings* [1] přeložil Jan Šustek.

```

1 \makeatletter
2 \def\otaznik{%
3   \@ifnextchar ?{\@otaznikS}{\@otaznikBez}}
4 \def\@otaznikS#1#2#3{S~otazníkem (#2) a (#3).}
5 \def\@otaznikBez#1#2{Bez otazníku (#1) a (#2).}
6 \makeatother

```

Profesor Lagally navrhoval k odstranění otazníku namísto řádku 4 psát

```
7 \def\@otaznikS?#1#2{S~otazníkem (#1) a (#2).}
```

V obou případech po použití

```

8 \otaznik?{první}{druhý}
9 \otaznik{první}{druhý}

```

dostaneme výsledek

S otazníkem (první) a (druhý).

Bez otazníku (první) a (druhý).

Child! do not throw this book about!

Refrain from the unholy pleasure

Of cutting all the pictures out!

Preserve it as your chiefest treasure!

A Bad Child's Book of Beasts

HILLAIRE BELLOC

2. METAPOST a pdfL^AT_EX

Program METAPOST slouží k vytváření PostScriptových obrázků. Tyto obrázky můžeme jednoduše vložit do L^AT_EXového dokumentu s výstupem do formátu dvi. Naproti tomu pdfL^AT_EX obecně neumí zpracovat PostScriptové soubory. Naštěstí umí zpracovat ty PostScriptové soubory, které byly vytvořeny METAPOSTem, a proto METAPOSTové obrázky mohou být přímo vloženy do dokumentu vytvořeného pdfL^AT_EXem. Toto však není tak přímočaré, jak by to mohlo být.

Mějme soubor s názvem `obr.mp`, který obsahuje řekněme tři obrázky. METAPOST vytvoří tři soubory `obr.1`, `obr.2` a `obr.3`, pro každý obrázek jeden soubor. L^AT_EXový balíček `graphicx` však očekává,¹ že PostScriptové soubory vytvořené

¹Autor článku očekává, že pro vkládání obrázků používáme makro `\includegraphics` balíčku `graphicx`. Další možností použití METAPOSTu je přeložení souboru `obr.mp` programem `mptopdf` namísto programu `mpost`, čímž se vytvoří soubory `obr-1.pdf`, `obr-2.pdf` a `obr-3.pdf`. Alternativní způsob vkládání METAPOSTových obrázků je pomocí makra `\convertMPtoPDF`, jak je vidět na stránce 63 tohoto čísla Zpravodaje, nebo přímé vložení výše zmíněných pdf souborů pomocí příkazů `\pdfximage`, `\pdfrefximage` a `\pdflastximage` popsané v též článku. (pozn. překl.)

METAPOSTem mají příponu `.mps`. Balíček můžeme následujícím způsobem nastavit tak, že se k souborům s neznámou příponou bude chovat, jako by měly příponu `.mps`.

```
10 \DeclareGraphicsRule{*}{mps}{*}{}

```

L^AT_EX (na rozdíl od pdfL^AT_EXu), nebo přesněji programy jako `dvips` nebo `xdvi`, umí zpracovat zapouzdřené PostScriptové soubory (s příponou `.eps`). Balíček `graphicx` můžeme nastavit podobným způsobem.

```
11 \usepackage{ifpdf}
12 \ifpdf
13   \usepackage{graphicx}
14   \DeclareGraphicsRule{*}{mps}{*}{}
15 \else
16   \usepackage{graphicx}
17   \DeclareGraphicsRule{*}{eps}{*}{}
18 \fi

```

Pokud chceme, aby obrázek METAPOSTu mohl být použitý v L^AT_EXu, potom na začátku METAPOSTového souboru napíšeme

```
19 prologues := 1;

```

To zařídí, že METAPOST vytvoří zapouzdřené PostScriptové soubory. Zdá se, že bez problémů můžeme nastavit stejné `prologues` také pro použití v pdfL^AT_EXu.²

A mathematician, like a painter or a poet, is a maker of patterns. If his patterns are more permanent than theirs, it is because they are made with ideas.

A Mathematician's Apology

G. H. HARDY

3. Spidrony

Jednoho dne jsem si slepě procházel *Science News*, když jsem si všiml krátkého článku Ivarse Petersona o spidronech [3]. Zájemci si mohou vygooglit „spidron“ pro více obrázků a informací o spidronech.

²Nastavení hodnoty `prologues` souvisí s vkládáním fontů do výsledného souboru s obrázkem. Aktuální verze METAPOSTu umožňuje nastavit různé hodnoty `prologues` pro různá použití. Pro podrobnosti vizte manuál k METAPOSTu. Překladatel během své 15leté práce v METAPOSTu při vkládání tisícovek obrázků makrem `\convertMPtoPDF` do pdf(L^A)T_EXu zatím hodnotu `prologues` nenastavoval a zatím nenarazil na problém. Proto si nemyslí, že by se měli začínající L^AT_EXisté nastavováním hodnoty `prologues` nějak trápit. (pozn. překl.)

Spidrony objevil a pojmenoval maďarský návrhář a grafik Dániel Erdély, když si pohrával se šestiúhelníky. Spidrony jsou tvořeny stále menšími navazujícími rovnostrannými a rovnoramennými trojúhelníky.

Uvědomil jsem si, že spidrony můžu nakreslit v METAPOSTu a po několika pokusech jsem napsal následující METAPOSTová makra pro kreslení spidronů.

```

20 %% soubor semispid.mp
21 % MetaPostové makro k nakreslení semispidronu
22 % semispid(střed, vrchol, iterace, barva1, barva2, směr)
23 def semispid(suffix $$, $)%
24     (expr iter, shadea, shadeb, clock) =
25     if clock: hxa := -60; else: hxa := 60; fi
26     pair v[];
27     path phex[];
28     v0 := z$$;
29     v1 := z$;
30     % vnější šestiúhelník
31     for i := 2 upto 6:
32         v[i] := v1 rotated around (v0, (i-1)*hxa);
33     endfor
34     z$a = v1; z$b = v2; z$c = v3;
35     z$d = v4; z$e = v5; z$f = v6;
36     phex0 := v1--v2--v3--v4--v5--v6--cycle;
37     if showverts:
38         dotlabels.lft($a,$e,$f);
39         dotlabels.rt($b,$c,$d);
40     fi
41     if showlines:
42         draw v1--v3--v5--cycle;
43         draw v2--v4--v6--cycle;
44     fi
45     % konstrukce trojúhelníků
46     for n:= 1 upto iter:
47         k := 10(n-1);
48         j := 10n;
49         v[1+j] := (v[1+k]--v[3+k])
50                 intersectionpoint
51                 (v[2+k]--v[6+k]);
52         for i := (2+j) upto (6+j):
53             v[i] := v[1+j]
54                     rotated around
55                     (v0, (i-1-j)*hxa);
56     endfor

```

```

57   if showlines:
58       draw v[1+j]--v[3+j]--v[5+j]--cycle;
59       draw v[2+j]--v[4+j]--v[6+j]--cycle;
60   fi
61   phex[n] := v[1+j]--v[1+k]--v[2+k]--cycle;
62   phex[n+1] := v[1+j]--v[2+j]--v[2+k]--cycle;
63   fill phex[n] withcolor shadea;
64   fill phex[n+1] withcolor shadeb;
65   if showcells:
66       draw phex[n]; draw phex[n+1];
67   fi
68   if showedges:
69       draw v[1+k]--v[1+j];
70       draw v[2+k]--v[2+j];
71   fi
72   endfor
73   if showedges: draw v[1+j]--v[2+j]; fi
74   if showhex: draw phex0; fi
75   enddef;

```


Jak název napovídá, makro **semispid** slouží k nakreslení poloviny spidronu – tu Erdély pojmenoval semispidron. Konstruovaný semispidron je umístěn uvnitř šestiúhelníku. Makru jsou předány pozice středu tohoto šestiúhelníku a jednoho z jeho vrcholů. Další argumenty určují počet trojúhelníků a barvy použité pro jednotlivé trojúhelníky. Makro používá booleovské proměnné definované jinde a podle nich zobrazuje různé části konstrukce obrázků.

K nakreslení obrázku 1 jsem použil následující program.

```

76 %% soubor glstr9.mp
77 % obrázky spidronů
78 prologues := 1;
79 input semispid
80 %% deklarace a počáteční nastavení booleovských proměnných
81 % zobrazení základního šestiúhelníku
82 boolean showhex; showhex := false;
83 % popisky vrcholů
84 boolean showverts; showverts := false;
85 % nakreslení pomocných čar
86 boolean showlines; showlines := false;
87 % nakreslení obrysů trojúhelníků
88 boolean showcells; showcells := false;
89 % otáčení po směru ručiček (ano = true)
90 boolean rh; rh := false;

```




Obrázek 1: Spidron

```

91 % nakreslení obrysu spidronu
92 boolean showedges; showedges := false;
93 % nastavení barev
94 color light,dark;
95 light := 0.1[white,black];
96 dark := 0.2[white,black];
97 beginfig(1); % spidron
98 u := 1in; % základní jednotka
99 showhex := false;
100 showverts := false;
101 showlines := false;
102 showcells := false;
103 rh := false;
104 showedges := false;
105 % střed a počáteční vrchol
106 z0 = (0,0);
107 z1 = (x0-2u,y0) rotatedaround(z0,60);
108 semispid(0, 1, 9, dark, light, rh);
109 y0-y1a = y1a-y10; x10=x0;
110 z11 = z1b;
111 semispid(10, 11, 9, light, dark, rh);
112 endfig;
113 %% další obrázky od řádku 118
114 end

```




Obrázek 2: Detaily konstrukce spidronu

Detaily konstrukce semispidronu jsou ukázány na obrázku 2. Makro **semispid** vygeneruje vrcholy šestiúhelníku a označí je **a** až **f**, kde **a** je zadaný vrchol, který je argumentem makra. Šestiúhelník je spojnicemi svých vrcholů rozdělen na trojúhelníky a na menší šestiúhelník. Podobně je tento šestiúhelník rozdělen na menší a takto pokračujeme tak dlouho, dokud vzniklý šestiúhelník není dostatečně malý. Vybarvené šestiúhelníky tvoří semispidron začínající na straně **ab** a končící blízko středu původního šestiúhelníku. Druhá polovina spidronu vznikne otočením kolem středu strany **ab** o 180 stupňů a přehozením barev.


Spidrony jsou obrazce, s nimiž je možné pokrýt rovinu. Návod na toto pokrytí je na obrázku 3, kde jsou tři semispidrony vepsané do šestiúhelníku. Prázdná místa je možné vyplnit dalšími třemi semispidrony. A celou rovinu je možné pokrýt šestiúhelníky. Odtud již vyplývá, že je rovinu možné pokrýt spidrony. Pokud vhodně budeme měnit barvy spidronů, dostaneme zajímavé vzory, jako například na obrázku 4. Mnohem více způsobů pokrytí roviny najdete v knize [4] – spidrony tam však nenajdete, protože v době vydání knihy ješně nebyly známy.

Napojením dvou semispidronů může vzniknout ještě jiný obrazec. Ve spidronu je jeden semispidron oproti druhému otočený. Pokud semispidrony napojíme proti sobě zrcadlově, dostaneme obrazec, který Erdély pojmenoval *hornflake*³. Můžete


³Překladatelovi se nepodařilo dohledat původní Erdélyovo pojmenování tohoto útvaru, proto použil překlad názvu uvedený v [1]. Dle jeho názoru se pouhým převzetím překladu názvu dopustíme menší nepřesnosti než doslovným dvojnásobným překladem z neznámého původního názvu. (pozn. překl.)




Obrázek 3: Tři semispidrony v šestiúhelníku



Obrázek 4: Pokrytí roviny – pouze spidrony



Obrázek 5: Hornflake



Obrázek 6: Pokrytí roviny – hornflaky a spidrony

jej vidět na obrázku 5. Na rozdíl od spidronů nejsou hornflaky obrazce, s nimiž je možné pokrýt rovinu. Když je však vhodně zkombinujeme se spidrony, rovinu můžeme pokrýt, jak je vidět na obrázku 6.

Ve svém článku Peterson píše:

Erdély měl na papíře namalované šestiúhelníky, rozmístěné, jako by to byly kachličky v koupelně. Pak začal papír přehýbat nahoru a dolů na hranách jednotlivých spidronů, přičemž ve středu šestiúhelníků nechával malé dírky. Takto pomačkal celý papír, až z obyčejných šestiúhelníků udělal dramatický trojrozměrný reliéf.

Ukazuje se, že ze spidronových vzorů je možné vytvořit nové trojrozměrné útvary podobné krystalům se spirálovitě uspořádanými stěnami.


Co však v onom článku chybí, je nějaký náznak, jaké jsou ty správné kombinace přehybů, pomocí kterých docílíme uvedené efekty. Po delším vyhledávání na internetu jsem našel následující Erdélyovy poznámky [5].

V dané soustavě spidronů jsem každou druhou hranu směřující do středu vzniklého šestiúhelníku přehnul směrem dolů (vznikly přehyby tvaru kopce) a zbývající hrany směrem nahoru (vznikly přehyby tvaru údolí). Vzniklý reliéf pod vlivem vnějších deformací nevykazuje jednoduché lineární vzory, jako například harmonika, ale hrany reliéfu proudí od vrcholů šestiúhelníků do jejich středů jako vír.

Po dlouhém přemýšlení a experimentování jsem došel k závěru, že jedna z těch správných kombinací přehybů je ta, která je na obrázku 7, na kterém je polovina šestiúhelníku a tři semispidrony. Tečkované čáry označují přehyby nahoru (údolí) a plné čáry označují přehyby dolů (kopce).

Pokud si chcete vytvořit obrovskou kresbu a zkoušet přehyby, tady je zdrojový kód obrázku 4, se kterým si můžete hrát. Samozřejmě si zdrojový kód můžete upravit podle libosti.

```
115 %% soubor glstr9.mp
116 % obrázky spidronů
117 % předchozí obrázky po řádek 112
118 beginfig(5); % pokrytí roviny spidrony
119 u := 0.175in;
120 showhex := false;
121 showverts := false;
122 showlines := false;
123 showcells := false;
124 rh := false;
125 showedges := false; showedges := true;
126 color cola, colb;
127 cola := light; colb := dark;
128 depth := 7;
```



Obrázek 7: Přehyby

```

129 rad := 2u;
130 z0 = (0,0);
131 % základní šestiúhelník
132 for kn := 1 upto 6:
133 z[kn] = (x0-2u,y0) rotatedaround(z0,60*(kn-1));
134 if odd kn:
135   cola := light;
136 else:
137   cola := dark;
138 fi
139 colb := cola;
140 semispid(0, [kn], depth, cola, colb, rh);
141 endfor
142 % kopie původního šestiúhelníku tvořící pokrytí
143 shd := (sqrt 3)/2*rad; % posun nahoru/dolů
144 shr := 3rad;           % posun doleva/doprava

```

```

145 picture pic[];
146 pic100 := currentpicture;
147 pic0 := pic100 shifted (0,2shd);
148 for kn := 1 upto 6:
149   pic[kn] := pic0 rotatedaround(z0,60kn);
150   draw pic[kn];
151 endfor
152 pic10 = pic100 shifted (0,4shd);
153 for kn := 1 upto 6:
154   pic[10+kn] := pic10 rotatedaround(z0,60kn);
155   draw pic[10+kn];
156 endfor
157 pic20 = pic100 shifted (3rad, 0);
158 for kn := 1 upto 6:
159   pic[20+kn] := pic20 rotatedaround(z0,60kn);
160   draw pic[20+kn];
161 endfor
162 endfig;
163 % další obrázky
164 end

```

Sám jsem to zkoušel a zjistil jsem, že je velmi složité správně přehnout i jen jeden šestiúhelník vytištěný na celou stránku A4. Proto jsem se rozhodl, že si vždy vezmu jeden spidron, ten přehnu a pak jednotlivé spidrony slepím k sobě lepicí páskou. Nakonec jsem své snažení završil tak, že jsem si raději na internetu prohlížel obrázky spidronových reliéfů, které vytvořili jiní (i když mám podezření, že většina těchto obrázků byla vytvořena na počítači a nebyla vytvořena ručně přehýbáním spidronů a vyfotografováním).

Odkazy

1. WILSON, Peter. Glisterings. *TUGboat*. 2009, roč. 30, č. 1, s. 69–73.
2. WILSON, Peter. Glisterings. *TUGboat*. 2008, roč. 29, č. 2, s. 324–327.
3. PETERSON, Ivars. Swirling seas, crystal balls. *Science News*. 2006, roč. 170, č. 17, s. 266–268.
4. GRÜNBAUM, Branko; SHEPHARD, Geoffrey Colin. *Tilings and Patterns*. W. H. Freeman, 1987.
5. ERDÉLY, Dániel. *Spidron system: A flexible space-filling structure*. 1998. Myšlenka z roku 1979, poprvé prezentováno na 12th International Conference on Crystal Growth (ICCG 1998).

Summary: It Might Work VIII – Drawings

Some aspects of inserting METAPOST figures into (pdf)L^AT_EX documents are discussed in this paper. The paper also focuses on shapes called spidrons and on manipulations with them.

Keywords: L^AT_EX, METAPOST, spidron

*Peter Wilson, herries.press@earthlink.net
18912 8th Ave. SW
Normandy Park, WA 98166 USA*